$P(x) = hx^3 - x^2 + (a+b)x - ab$ admet 3 racines dans $\mathbb{C}$ telles que $x_1 + x_2 + x_3 = 1/h, x_1x_2 + x_2x_3 + x_3x_1 = a+b, x_1x_2x_3 = ab$. $P(0) < 0, P(a) = ha^3 > 0, P(b) > 0, P(a+b) = -ab + h(a+b)^3 < 0$ dès que $h < \dfrac{ab}{(a+b)^3}$ et $\lim\limits_{+\infty} P = +\infty$ donc si $0 < h < \dfrac{ab}{(a+b)^3}$, les 3 racines sont réelles, avec $x_1 \in]0, a], x_2 \in [b, a+b[, x_3 \in]a+b, +\infty[$.
f est $\mathcal{C}^1$ sur $\mathbb{R}^2$ et $f(a,0) = 0$, $\dfrac{\partial f}{\partial x}(a,0) = b - a \neq 0$ donc sur un $I \times J$ tel que $(a,0) \in I \times J, J =]-\alpha, \beta[$, $f(x,h) = 0 \iff x = x_1(h)$ où x_1 est $\mathcal{C}^1(J)$. $x_1(0) = a$ et $x_1'(h) = -\dfrac{\partial f}{\partial h} / \dfrac{\partial f}{\partial x}$ et par récurrence x_1 est $\mathcal{C}^k$ pour tout k d'où existence et forme du DL.
On a de même $x_2(h) = b + ... + o(h^4)$ donc $x_3(h) = 1/h - x_1(h) - x_2(h) = 1/h - (a+b) - ... + o(h^4)$ ce qui est un développement asymptotique et x_3 est $\mathcal{C}^\infty$ sur $]0, \beta[$.
Le changement de variable proposé donne $f(x,h) = h(x - x_1)(x - x_2)(x - x_3) = h(x_3 - x)\cos^2 t((x_2 - x_1)/2)^2$ donc l'intégrale devient $\displaystyle\int_{-\pi/2}^{\pi/2} \frac{\mathrm{d}t}{\sqrt{h(x_3 - x)}}$ (intégrale elliptique non calculable avec les fonctions élémentaires)

O19-078

```
> restart;
>
> f:=h->(x-a)*(b-x)+h*x^3;
```

$$f := h \to (x-a)\,(b-x) + h\,x^3$$

```
> a:=1:b:=2:
> plot([f(0.01),f(0.02),f(0.05)],x=0..10,y=-10..1);
```




```
> a:='a':b:='b':p:=a+c[1]*h+c[2]*h^2+c[3]*h^3+c[4]*h^4;
```

$$p := a + c_1\,h + c_2\,h^2 + c_3\,h^3 + c_4\,h^4$$

```
> subs(x=p,f(h));
```

$$(c_4\,h^4 + c_3\,h^3 + c_2\,h^2 + c_1\,h + a)^3\,h + (c_4\,h^4 + c_3\,h^3 + c_2\,h^2 + c_1\,h)\,(-c_4\,h^4 - c_3\,h^3 - c_2\,h^2 - c_1\,h + b - a)$$

```
> eq:=expand(subs(x=p,f(h)));
```

$$eq := -c_1\,h\,a+3\,c_4^{\;2}\,h^{11}\,c_2-2\,c_4\,h^7\,c_3-2\,c_4\,h^6\,c_2-2\,c_4\,h^5\,c_1-2\,c_3\,h^5\,c_2-2\,c_3\,h^4\,c_1$$
$$-2\,c_2\,h^3\,c_1-c_4\,h^4\,a-c_3\,h^3\,a-c_2\,h^2\,a-c_4^{\;2}\,h^8-c_3^{\;2}\,h^6-c_2^{\;2}\,h^4-c_1^{\;2}\,h^2+c_4^{\;3}\,h^{13}+c_3^{\;3}\,h^{10}$$
$$+c_2^{\;3}\,h^7+c_1^{\;3}\,h^4+h\,a^3+3\,c_4^{\;2}\,h^{12}\,c_3+6\,c_4\,h^6\,c_1\,a+6\,c_4\,h^{10}\,c_3\,c_2+6\,c_4\,h^9\,c_3\,c_1+6\,c_4\,h^8\,c_2\,c_1$$
$$+6\,c_4\,h^8\,c_3\,a+6\,c_4\,h^7\,c_2\,a+6\,c_3\,h^5\,c_1\,a+6\,c_3\,h^7\,c_2\,c_1+6\,c_3\,h^6\,c_2\,a+6\,c_2\,h^4\,c_1\,a$$
$$+3\,c_4^{\;2}\,h^{10}\,c_1+3\,c_4^{\;2}\,h^9\,a+3\,c_4\,h^{11}\,c_3^{\;2}+3\,c_4\,h^9\,c_2^{\;2}+3\,c_4\,h^7\,c_1^{\;2}+3\,c_4\,h^5\,a^2+3\,c_3^{\;2}\,h^9\,c_2$$
$$+3\,c_3^{\;2}\,h^8\,c_1+3\,c_3^{\;2}\,h^7\,a+3\,c_3\,h^8\,c_2^{\;2}+3\,c_3\,h^6\,c_1^{\;2}+3\,c_3\,h^4\,a^2+3\,c_2^{\;2}\,h^6\,c_1+3\,c_2^{\;2}\,h^5\,a$$
$$+3\,c_2\,h^5\,c_1^{\;2}+3\,c_2\,h^3\,a^2+3\,c_1^{\;2}\,h^3\,a+3\,c_1\,h^2\,a^2+c_4\,h^4\,b+c_3\,h^3\,b+c_2\,h^2\,b+c_1\,h\,b$$

```
> sys:={seq(coeff(eq,h,i),i=1..4)};
```

$$sys := \{3\,c_1\,a^2-c_1^{\;2}-c_2\,a+c_2\,b, -c_1\,a+a^3+c_1\,b, c_3\,b-c_3\,a+3\,c_2\,a^2-2\,c_2\,c_1+3\,c_1^{\;2}\,a,$$
$$-c_2^{\;2}+3\,c_3\,a^2+c_4\,b+6\,c_2\,c_1\,a+c_1^{\;3}-c_4\,a-2\,c_3\,c_1\}$$

```
> sol:=solve(sys,{c[1],c[2],c[3],c[4]});
```

$$sol := \left\{c_3=\frac{a^7\,(5\,a^2-15\,a\,b+12\,b^2)}{a^5-5\,a^4\,b+10\,a^3\,b^2-10\,a^2\,b^3+5\,b^4\,a-b^5}, c_2=\frac{a^5\,(2\,a-3\,b)}{a^3-3\,a^2\,b+3\,a\,b^2-b^3},\right.$$
$$\left.c_1=\frac{a^3}{a-b}, c_4=\frac{a^9\,(14\,a^3-63\,a^2\,b+99\,a\,b^2-55\,b^3)}{a^7-7\,a^6\,b+21\,a^5\,b^2-35\,a^4\,b^3+35\,a^3\,b^4-21\,a^2\,b^5+7\,b^6\,a-b^7}\right\}$$

```
> pp:=subs(sol,p);
```

$$pp := \frac{a^9\,(14\,a^3-63\,a^2\,b+99\,a\,b^2-55\,b^3)\,h^4}{a^7-7\,a^6\,b+21\,a^5\,b^2-35\,a^4\,b^3+35\,a^3\,b^4-21\,a^2\,b^5+7\,b^6\,a-b^7}$$
$$+\frac{a^7\,(5\,a^2-15\,a\,b+12\,b^2)\,h^3}{a^5-5\,a^4\,b+10\,a^3\,b^2-10\,a^2\,b^3+5\,b^4\,a-b^5}+\frac{a^5\,(2\,a-3\,b)\,h^2}{a^3-3\,a^2\,b+3\,a\,b^2-b^3}+\frac{a^3\,h}{a-b}+a$$

```
> a:=1:b:=2:plot(pp,h=0..0.1);
```




```
> a:='a':b:='b':q:=1/h+cc[1]+cc[2]*h+cc[3]*h^2;
```

$$q := \frac{1}{h} + cc_1 + cc_2\,h + cc_3\,h^2$$

```
> h^2*subs(x=q,f(h));
```

$$h^2 \left(\left(\frac{1}{h} + cc_1 + cc_2\,h + cc_3\,h^2 - a \right) \left(b - \frac{1}{h} - cc_1 - cc_2\,h - cc_3\,h^2 \right) + h \left(\frac{1}{h} + cc_1 + cc_2\,h + cc_3\,h^2 \right)^3 \right)$$

```
> eqq:=expand(h^2*subs(x=q,f(h)));
```

$$eqq := 6\,h^6\,cc_1\,cc_2\,cc_3 - h^2\,a\,b + h^2\,cc_1\,b + h^2\,a\,cc_1 + h^3\,cc_2\,b + h^4\,cc_3\,b + h^3\,a\,cc_2 + h^4\,a\,cc_3 + 3\,h^4\,cc_1^{\,2}\,cc_2 + 3\,h^5\,cc_1^{\,2}\,cc_3 + 3\,h^7\,cc_2^{\,2}\,cc_3 + 3\,h^8\,cc_2\,cc_3^{\,2} + 3\,h^5\,cc_1\,cc_2^{\,2} + 3\,h^7\,cc_1\,cc_3^{\,2} + 4\,h^5\,cc_2\,cc_3 + 4\,h^3\,cc_1\,cc_2 + 4\,h^4\,cc_1\,cc_3 + h\,b + h\,a + h^2\,cc_2 + h\,cc_1 + h^3\,cc_3 + 2\,h^2\,cc_1^{\,2}$$

$$+2\,h^4\,{cc_2}^2+2\,h^6\,{cc_3}^2+h^6\,{cc_2}^3+h^9\,{cc_3}^3+h^3\,{cc_1}^3$$

```
> ssys:={seq(coeff(eqq,h,i),i=1..3)};
```

$$ssys := \{-a\,b+cc_1\,b+a\,cc_1+cc_2+2\,{cc_1}^2,\ cc_2\,b+a\,cc_2+4\,cc_1\,cc_2+cc_3+{cc_1}^3,\ b+a+cc_1\}$$

```
> ssol:=solve(ssys,{cc[1],cc[2],cc[3]});
```

$$ssol := \{cc_2=-a\,b-b^2-a^2,\ cc_3=-3\,a\,b^2-2\,b^3-3\,a^2\,b-2\,a^3,\ cc_1=-b-a\}$$

```
> qq:=subs(ssol,q);
```

$$qq := \frac{1}{h}-b-a+(-a\,b-b^2-a^2)\,h+(-3\,a\,b^2-2\,b^3-3\,a^2\,b-2\,a^3)\,h^2$$

```
> assume(x[1]<x[2],x[2]<x[3]):i:=simplify(int(1/sqrt(h*(x[3]-(x[1]
+x[2])/2-(x[2]-x[1])*sin(t)/2)),t=-Pi/2..Pi/2));
```

$$i := -\frac{2\,\mathrm{EllipticK}\left(\frac{\sqrt{x\sim_2-x\sim_1}}{\sqrt{-x\sim_1+x\sim_3}}\right)}{\sqrt{h}\,\sqrt{-x\sim_1+x\sim_3}}$$

```
>
```